Import('*')

env = env.Clone()

env.Prepend(LIBS = [mesautil, gallium])

if env['platform'] in ('freebsd8', 'sunos'):
    env.Append(LIBS = ['m'])

if env['platform'] == 'freebsd8':
    env.Append(LIBS = ['pthread'])

progs = [
    'pipe_barrier_test',
    'u_cache_test',
    'u_half_test',
    'translate_test'
]

for progname in progs:
    prog = env.Program(
        target = progname,
        source = progname + '.c',
    )
    if progname not in [
        'u_cache_test', # too long
        'translate_test', # unreliable
    ]:
       env.UnitTest(progname, prog)
